


# CSCE 611

## Testbenches, Simulation, and JTAG and the MIPS ALU



Instructor: Jason D. Bakos



---

# Tutorial Outline


---

- Tutorial 1: Simple adder
  - Design register and full adder using Verilog
  - Connect full adders to create 9-bit adder using Quartus schematic tool
  - Create test bench to verify functionality
  - Synthesize and implement on DE2 board
  - Convert to carry lookahead adder to improve clock speed
  - Convert to native adder to compare clock speed
- Tutorial 2: MIPS ALU
  - Design ALU that we'll use in the CPU designer later
  - Implement all MIPS arithmetic, logic, shift, and comparison instructions



# Review: 1-Bit Adders

## Half Adder




A	B	$C_{out}$	S
0	0	0	0
0	1	0	1
1	0	0	1
1	1	1	0

$$S = A \oplus B$$

$$C_{out} = AB$$

## Full Adder



$C_{in}$	A	B	$C_{out}$	S
0	0	0	0	0
0	0	1	0	1
0	1	0	0	1
0	1	1	1	0
1	0	0	0	1
1	0	1	1	0
1	1	0	1	0
1	1	1	1	1


$$S = A \oplus B \oplus C_{in}$$

$$C_{out} = AB + AC_{in} + BC_{in}$$



# Review: Ripple-Carry Adder

- Chain 1-bit adders together
- Carry ripples through entire chain
- Disadvantage: **slow**



---

# Full Adder

---

- File | New | Verilog HDL File

```
module full_adder (input a,b,cin,output s,cout);  
    assign s = a ^ b ^ cin;  
    assign cout = (a & b) | (a & cin) | (b & cin);  
endmodule
```

Save as "fulladder.v"



---

# Bottom-up Design

---

- Combine 9 full adders to create 9 bit adder
- Create new Verilog file:

```
module adder (input [8:0] a,b,output [8:0] s,output cout);  
    wire [9:0] carry_chain;  
  
    full_adder add0(a[0],b[0],1'b0,s[0],carry_chain[0]);  
    full_adder add1(a[1],b[1],carry_chain[0],s[1],carry_chain[1]);  
    full_adder add2(a[2],b[2],carry_chain[1],s[2],carry_chain[2]);  
    full_adder add3(a[3],b[3],carry_chain[2],s[3],carry_chain[3]);  
    full_adder add4(a[4],b[4],carry_chain[3],s[4],carry_chain[4]);  
    full_adder add5(a[5],b[5],carry_chain[4],s[5],carry_chain[5]);  
    full_adder add6(a[6],b[6],carry_chain[5],s[6],carry_chain[6]);  
    full_adder add7(a[7],b[7],carry_chain[6],s[7],carry_chain[7]);  
    full_adder add8(a[8],b[8],carry_chain[7],s[8],carry_chain[8]);  
  
    assign cout = carry_chain[8];  
  
endmodule
```

- Save as "adder.v"



---

# Bottom-up Design

---

- Copying and pasting each instantiation is bulky and isn't amenable to variable adder sizes
- Use generate statement:

```
module adder (input [8:0] a,b,output [8:0] s,output cout);  
    wire [9:0] carry_chain;  
    genvar i;  
  
    generate for (i=0;i<9;i=i+1) begin : adder_array  
        if (i==0)  
            full_adder add(a[i],b[i],1'b0,s[i],carry_chain[i]);  
        else  
            full_adder add(a[i],b[i],carry_chain[i-1],s[i],carry_chain[i]);  
        end  
    endgenerate  
  
    assign cout = carry_chain[8];  
  
endmodule
```



---

# Parameters

---

- Now let's add a parameter:

```
module adder #(parameter width=32) (input [width-1:0] a,b,output [width-1:0] s,output cout);
    wire [width-1:0] carry_chain;
    genvar i;
    generate for (i=0;i<width;i=i+1) begin : adder_array
        if (i==0)
            full_adder add(a[i],b[i],1'b0,s[i],carry_chain[i]);
        else
            full_adder add(a[i],b[i],carry_chain[i-1],s[i],carry_chain[i]);
        end
    endgenerate
    assign cout = carry_chain[width-1];
endmodule
```





---

# Register

---

- Let's create a register
- New Verilog file:

```
module regn #(parameter width=32) (input clk,input [width-1:0]  
d,output reg [width-1:0] q);
```

```
always @(posedge clk)  
begin  
    q <= d;  
end
```

```
endmodule
```

- Save as regn.v

# Adder Datapath


- Create an adder datapath, where registers supply input data and registers capture output data:

```
module adder_reg #(parameter width=32) (input clk,input [width-1:0] a,b,output[width-1:0] s,output cout);
```

```
    wire [width-1:0] a_reg,b_reg,s_comb;  
    wire cout_comb;
```

```
    regn #(width) input_reg_a(clk,a,a_reg);  
    regn #(width) input_reg_b(clk,b,b_reg);  
    adder#(width) my_adder(a_reg,b_reg,s_comb,cout_comb);  
    regn #(width) output_reg_s(clk,s_comb,s);  
    regn #(1) output_reg_cout(clk,cout_comb,cout);
```

```
endmodule
```



---

# Testbenches

---

- HDL code written to test another HDL module, the *device under test* (dut), also called the *unit under test* (uut)
- Generally not synthesizable
- Types of testbenches:
  - Simple testbench
  - Self-checking testbench
  - Self-checking testbench with testvectors

---

# Example

---

```
module sillyfunction(input  a, b, c,  
                      output y);  
    assign y = ~b & ~c | a & ~b;  
endmodule
```



---

# Simple Testbench

---

```
module testbench1();  
    reg  a, b, c;  
    wire y;  
    // instantiate device under test  
    sillyfunction dut(a, b, c, y);  
    // apply inputs one at a time  
    initial begin  
        a = 1'b0; b = 1'b0; c = 1'b0; #10;  
        c = 1'b1; #10;  
        b = 1'b1; c = 1'b0; #10;  
        c = 1'b1; #10;  
        a = 1'b1; b = 1'b0; c = 1'b0; #10;  
        c = 1'b1; #10;  
        b = 1'b1; c = 1'b0; #10;  
        c = 1'b1; #10;  
    end  
endmodule
```



# Self-checking Testbench

```
module testbench2();
    reg a, b, c;
    wire y;
    // instantiate device under test
    sillyfunction dut(a, b, c, y);
    // apply inputs one at a time
    // checking results
    initial begin
        a = 1'b0; b = 1'b0; c = 1'b0; #10;
        if (y !== 1'b1) $display("000 failed.");
        c = 1'b1; #10;
        if (y !== 1'b0) $display("001 failed.");
        b = 1'b1; c = 1'b0; #10;
        if (y !== 1'b0) $display("010 failed.");
        c = 1'b1; #10;
        if (y !== 0) $display("011 failed.");
        a = 1'b1; b = 1'b0; c = 1'b0; #10;
        if (y !== 1'b1) $display("100 failed.");
        c = 1'b1; #10;
        if (y !== 1) $display("101 failed.");
        b = 1'b1; c = 1'b0; #10;
        if (y !== 1'b0) $display("110 failed.");
        c = 1'b1; #10;
        if (y !== 1'b0) $display("111 failed.");
    end
endmodule
```

---


# Testbench with Testvectors

---

- Write testvector file: inputs and expected outputs
- Testbench:
  1. Generate clock for assigning inputs, reading outputs
  2. Read testvectors file into array
  3. Assign inputs, expected outputs
  4. Compare outputs to expected outputs and report errors

# Testbench with Testvectors

- Testbench clock is used to assign inputs (on the rising edge) and compare outputs with expected outputs (on the falling edge).



- The testbench clock may also be used as the clock source for synchronous sequential circuits





---

# Testvectors File

---

**File:** `example.tv` - contains vectors of `abc_yexpected`

`000_1`

`001_0`

`010_0`

`011_0`

`100_1`

`101_1`

`110_0`

`111_0`



---

# Testbench: 1. Generate Clock

---

```
module testbench3();  
    reg            clk, reset;  
    reg            a, b, c, yexpected;  
    wire           y;  
    reg [31:0] vectornum, errors;    // bookkeeping variables  
    reg [3:0]  testvectors[10000:0]; // array of testvectors  
  
    // instantiate device under test  
    sillyfunction dut(a, b, c, y);  
  
    // generate clock  
    always         // no sensitivity list, so it always executes  
        begin  
            clk = 1'b1; #5; clk = 1'b0; #5;  
        end
```



---

## 2. Read Testvectors into Array

---

```
// at start of test, load vectors
// and pulse reset
```

```
initial
begin
    $readmemb("example.tv", testvectors);
    vectornum = 0; errors = 0;
    reset = 1'b1; #27; reset = 1'b0;
end
```

```
// Note: $readmemb reads testvector files written in
// hexadecimal (as opposed to $readmemh in binary)
// Can insert Verilog comments into file
// Can also add addresses (use @ with hex address)
```



---

## 3. Assign Inputs and Expected Outputs

---

```
// apply test vectors on rising edge of clk
always @(posedge clk)
begin
    #1; {a, b, c, yexpected} = testvectors[vectornum];
end
```



---

## 4. Compare Outputs with Expected Outputs

---

```
// check results on falling edge of clk
always @(negedge clk)
  if (~reset) begin // skip during reset
    if (y !== yexpected) begin
      $display("Error: inputs = %b", {a, b, c});
      $display("  outputs = %b (%b expected)", y, yexpected);
      errors = errors + 1;
    end
  end

// Note: to print in hexadecimal, use %h. For example,
//       $display("Error: inputs = %h", {a, b, c});
```

---

## 4. Compare Outputs with Expected Outputs

---

```
// increment array index and read next testvector
    vectornum = vectornum + 1;
    if (testvectors[vectornum] === 4'bx) begin
        $display("%d tests completed with %d errors",
            vectornum, errors);
    $finish;
    end
end
endmodule
```

```
// Note: === and !== can compare values that are
// x or z.
```



# Another Testbench

```
module testbench();

reg clk, reset;
reg [31:0] vectornum, errors; // bookkeeping variables
reg [23:0] testvectors[10000:0]; // array of testvectors

reg [7:0]
a,b,sexpected,sexpected_dly1,sexpected_dly2,sexpected_dly3,a_dly1,a
_dly2,b_dly1,b_dly2;
wire [7:0] s;
wire cout;

adder_reg #(8) myadder (clk,a,b,s,cout);

always begin
    clk=1;#5;clk=0;#5;
end

initial begin
    $readmemh("addtests.tv", testvectors);
    vectornum = 0; errors = 0;
    reset = 1; #4; reset = 0;
end

always @(posedge clk) begin
    {a, b, expected} <= testvectors[vectornum];
    expected_dly1 <= expected;
    expected_dly2 <= expected_dly1;
    a_dly1 <= a;
    a_dly2 <= a_dly1;
    b_dly1 <= b;
    b_dly2 <= b_dly1;

    // check results on falling edge of clk
    always @(negedge clk) begin
        if (s !== expected_dly2) begin
            $display("Error: a = %d, b =
%d, s = %d, t = %d", a_dly2,b_dly2,s,$time);
            errors = errors + 1;
        end
        vectornum = vectornum + 1;
        if (testvectors[vectornum] === 4'bx) begin
            $display("%d tests completed
with %d errors",vectornum, errors);
            $finish;
        end
    end
endmodule
```




# Simulation

- In order to simulate we need to start Modelsim separately
- In terminal window:  
`cd quartus_work`  
`vsim&`
- Close any startup windows


1

2






# Modelsim




# Modelsim

- Select Compile | Compile...
- Select all five Verilog files for our design and click Compile then Done





# Modelsim

- Now select Simulate | Start Simulation
- Under the work library select testbench and click Ok



# Modelsim Windows



# Design Structure


The screenshot shows a simulation window titled "sim - Default". On the left is a hierarchical tree view of the design structure. On the right is a table with three columns: "Instance", "Design unit", and "Design unit".

Instance	Design unit	Design unit
testbench	testbench	Module
myadder	adder_reg	Module
input_reg_a	regn	Module
#ALWAYS#3	regn	Process
input_reg_b	regn	Module
#ALWAYS#3	regn	Process
my_adder	adder	Module
adder_array[8]	adder	VGenerat
adder_array[7]	adder	VGenerat
adder_array[6]	adder	VGenerat
adder_array[5]	adder	VGenerat
adder_array[4]	adder	VGenerat
adder_array[3]	adder	VGenerat
adder_array[2]	adder	VGenerat
adder_array[1]	adder	VGenerat
adder_array[0]	adder	VGenerat
#ASSIGN#13	adder	Process
output_reg_s	regn	Module
#ALWAYS#3	regn	Process
output_reg_cout	regn	Module
#ALWAYS#3	regn	Process
#ALWAYS#9	testbench	Process
#INITIAL#13	testbench	Process
#vsim_capacity#		Capacity





# Design Objects

- Note: the contents of this pane depend on what is selected in structure pane




# Wave Window



# Modelsim

- In transcript window, type "run 30000"



```
VSIM 69> run 30000
# TEST DATA
# 006 + 124 = 12a
# 19b + 000 = 19b
# 1d5 + 002 = 1d7
# 006 + 124 = 12a
# 19b + 000 = 19b
# 1d5 + 002 = 1d7
# 006 + 124 = 12a
# 19b + 000 = 19b
# 1d5 + 002 = 1d7
# 006 + 124 = 12a
# 19b + 000 = 19b
# 1d5 + 002 = 1d7
# 006 + 124 = 12a
# 19b + 000 = 19b
# 1d5 + 002 = 1d7
# 006 + 124 = 12a
# 19b + 000 = 19b
```





---

# Modelsim Notes

---

- Need to recompile after any changes
- “restart -f” will reload and restart simulation
- Signals must be added to wave or “log” in order to track
- You can add signals in lower levels of hierarchy by “drilling down” in structure and choosing signals in objects window
- Other useful Modelsim commands
  - force A X“DEADBEEF”
  - force op “0001”
  - force push 1
  - run 100
  - force -freeze clk 1 0, 0 {50 ps} -r 100

---

# Modelsim Notes

---

- Can also create a script for Modelsim:
- Example: test.do:



```
restart -f
force -freeze clk 1 0, 0 {50 ps} -r 100
force rst 1
force push 0
force pop 0
force data_in X"00000000"
run
force rst 0
force data_in X"00DECADE"
force push 1
run
force data_in X"00DEFACE"
run
```

```
force data_in X"0000FADE"
run
force data_in X"00DECAFE"
run
force pop 1
force push 0
run
run
run
run
force pop 0
run
```




# Synthesis

- Back to Quartus...
- Set adder\_reg as top-level entity
- Select device
  - Assignments | Settings
  - Click “Device”
  - Select EP2C35F672C6 and click OK




# Synthesis

- To compile design, click start button




- Once done, let's examine the FPGA resources required by the design




# Resource Usage

- Logic elements
  - Cyclone 2 LE contains one 4-input LUT and one 1-bit register
    - Usage: 119 out of 33,216
      - 22 LUT only
      - 41 register only
      - 56 both
- No multiplier usage (out of 70)
- No M4K usage (out of 105)



# Clock Speed



The screenshot shows the Xilinx ISE interface with the 'testbench.v' file open. The 'Table of Contents' pane on the left lists various reports, with 'Slow Model Fmax Summary' selected. The main window displays the 'Slow Model Fmax Summary' table, which provides timing information for the clock 'clk'.

	Fmax	Restricted Fmax	Clock Name	Note
1	81.31 MHz	81.31 MHz	clk	



---

# Adjust Adder

---

- Let's allow the synthesizer to build the adder architecture instead of us
  - Replace adder.v with:

```
module adder #(parameter width=32) (input [width-1:0] a,b,output [width-1:0] s,output
cout);

    wire [width:0] s_int;

    assign s_int = a + b;
    assign s = s_int[width-1:0];
    assign cout = s_int[width];

endmodule
```

- Re-simulate testbench to make sure it still works!
- New synthesis results:
  - 97 LEs (vs. 119)
  - 246.18 MHz (vs. 81.31 MHz!!!)



---

# JTAG

---

- Original idea: boundary scan
  - Originally used to test chip package and PCB connections
  - Chip inputs and outputs
  - “Scan chain”
- Now used as a way to source and probe internal signals
- Also used to program FPGAs
- “Joint Test Action Group” (JTAG) standard
  - Test clock (TCLK)
  - Test mode select (TMS)
  - Test data input (TDI)
  - Test data output (TDO)



---

# In System Testing

---

- Create a top-level wrapper, top.v:

```
module top (input CLOCK_50);
```

```
    wire [31:0] a,b,s;
```

```
    wire cout;
```

```
    adder_reg(.clk(CLOCK_50), .a(a), .b(b), .s(s), .cout(cout));
```

```
endmodule
```



---

# Sources and Probes

---

- Add sources and probes to the top-level:

```
altsource_probe #(.source_width(32),.instance_id("a")) input_a (.source(a),.probe());  
altsource_probe #(.source_width(32),.instance_id("b")) input_b (.source(b),.probe());  
altsource_probe #(.probe_width(32),.instance_id("s")) output_s (.source(),.probe(s));  
altsource_probe #(.probe_width(1),.instance_id("cout")) output_cout (.source(),.probe(cout));
```

- Specify your top-level design and implement on FPGA
- Go to Tools | In-System Sources and Probes Editor

# Sources and Probes

The screenshot displays the 'In-System Sources and Probes Editor' interface. At the top, a yellow banner indicates 'JTAG configuration invalid'. Below this, the 'Probe read interval' is set to 'Automatic' with a current interval of '0 samples per second'. The 'Event log' section shows a maximum size of '8' and a checked option to 'Save data to event log'. The 'Write source data' is set to 'Continuously'.

The main table lists sources and probes. The 'Sources' tab is active, showing a list of sources with columns for Index, Type, Alias, Name, and Data. The data is as follows:

Index	Type	Alias	Name	Data
P0			altera_reserved_unconnected_port	0
S31			source[31]	0
S30			source[30]	0
S29			source[29]	0
S28			source[28]	0
S27			source[27]	0
S26			source[26]	0
S25			source[25]	0
S24				
S23				
S22				
S21				
S20				
S19				
S18				
S17				
S16				
S15				
S14				
S13				
S12				
S11				
S10				
S9				
S8				
S7				
S6				
S5				
S4				
S3				
S2				
S1				
S0				

A context menu is open over the 'source[31]' entry, showing options: 'Undo', 'Select All', 'Delete Alias', 'Group', and 'Set Value of Source'. The 'Group' option is highlighted.

On the right, the 'JTAG Chain Configuration' panel shows 'Hardware: USB-Blaster [USB-0]' and 'Device: @1: EP3C120/EP4CE115 (0x020)'. The 'File' field is circled in red, showing the path 'C:\designs\adder\_path\output\_files\adder\_path.sof'.

# Sources and Probes

- Highlight each of the four sources/probes and click the "continuously read probe data" button:
- Enter values for A and B and watch the output value update in real time

Instance Manager: Ready to acquire

Probe read interval: Continuously Read Probe Data

Current interval: 9 samples per second

Index	Instance ID	Status	Sources: 66	Probes: 35	Name
0	a	Unloading data	32	1	altsource_probe:input_a
1	b	Unloading data	32	1	altsource_probe:input_b
2	s	Unloading data	1	32	altsource_probe:output_s
3	cout	Unloading data	1	1	altsource_probe:output_cout

**0 a**

Index	Type	Alias	Name	Data
P0			altera_reserved_unconnected_port	0
S[31..0]		source[31..0]		20

**1 b**

Index	Type	Alias	Name	Data
P0			altera_reserved_unconnected_port	0
S31		source[31]		0
S[30..0]		source[30..0]		10

**2 s**


Index	Type	Alias	Name	Data
P[31..0]			adder_reg:comb_3[s[31..0]	30
S0		source[0]		0

**3 cout**

Index	Type	Alias	Name	Data
P0			adder_reg:comb_3[cout]	0
S0		source[0]		0


# SignalTap

- Go to Tools | SignalTap II Logic Analyzer
- Set up sampling clock and SOF file:





# SignalTap

- Right-click in middle pane and select "Add Nodes"
- Type "\*a\*" and click "List"
- Hold down control to select all the q bits from the a-register
- Click the ">" button
- Click OK



# SignalTap

- Right-click newly-added signals and select "group"



# SignalTap

- Repeat process for b, s, and cout

The screenshot shows the SignalTap II Logic Analyzer interface. The top menu bar includes File, Edit, View, Project, Processing, Tools, Window, and Help. The Instance Manager shows a table with columns: Instance, Status, LEs, Memory, Small, Medium, and Large. The instance 'auto\_sigtap\_0' is listed with status 'Not running'. The JTAG Chain Configuration panel shows 'JTAG ready' and 'Hardware: USB-Blaster [USB-0]'. The Signal Configuration panel shows 'Clock: CLOCK\_50' and 'Sample depth: 128'. The Data table shows the following configuration:

Type	Alias	Name	Data Enable	Trigger Enable	Trigger Conditions
		...put_reg_a[q[0..31]	<input checked="" type="checkbox"/>	<input checked="" type="checkbox"/>	XXXXXXXXh
		...put_reg_b[q[0..31]	<input checked="" type="checkbox"/>	<input checked="" type="checkbox"/>	XXXXXXXXh
		...tput_reg_s[q[0..31]	<input checked="" type="checkbox"/>	<input checked="" type="checkbox"/>	XXXXXXXXh
		...n.output_reg_cout[q[0]	<input checked="" type="checkbox"/>	<input checked="" type="checkbox"/>	


The Hierarchy Display panel shows the project structure: top > adder\_reg:comb\_3 > regn:input\_reg\_a, regn:input\_reg\_b, regn:output\_reg\_s, regn:output\_reg\_cout. The Data Log panel shows the instance 'auto\_sigtap\_0'.





# SignalTap

- Set value of "FFFFFFFF" under trigger for A and click "run analysis"



trigger: 2016/08/24 12:24:09 #0			Lock mode: Allow all changes		
Node			Data Enable	Trigger Enable	Trigger Conditions
Type	Alias	Name	97	97	1 Basic
R		...n:input_reg_a[q[0..31]	<input checked="" type="checkbox"/>	<input checked="" type="checkbox"/>	FFFFFFFFh
R		...n:input_reg_b[q[0..31]	<input checked="" type="checkbox"/>	<input checked="" type="checkbox"/>	XXXXXXXXh
R		...:output_reg_s[q[0..31]	<input checked="" type="checkbox"/>	<input checked="" type="checkbox"/>	XXXXXXXXh
R		...regn:output_reg_cout[q[0]	<input checked="" type="checkbox"/>	<input checked="" type="checkbox"/>	



# SignalTap

- Back in "In System Sources and Probes", set A to -1

The left screenshot shows the 'In-System Sources and Probes Editor' window. It displays a table of sources and probes. The table has columns for Index, Instance ID, Status, Sources, and Probes. The table is as follows:

Index	Instance ID	Status	Sources	Probes
0	a	Not running	32	1
1	b	Not running	32	1
2	s	Not running	1	32
3	cout	Not running	1	1

The right screenshot shows the 'SignalTap' window. It displays a waveform display. A red circle and arrow highlight the value -1 in the 'Data' column of the table and the corresponding value in the waveform.

The bottom part of the left screenshot shows a table of data for instance 'a'. The table has columns for Index, Type, Alias, Name, and Data. The table is as follows:

Index	Type	Alias	Name	Data
P0			altera_reserved_unconnected_port	0
S[31..0]			source[31..0]	-1

The right screenshot shows a waveform display with a time axis from -8 to 22. The waveform shows a signal that is 0 for most of the time, but has a single pulse to -1 at time 0. A red circle and arrow highlight this pulse.




---

# ALU Design

---

- Let's create our MIPS ALU
- Requirements:
  - 32-bit
  - Implement 13 operations:
    - ADD, SUB, MULT, MULTU
    - AND, OR, NOR, XOR
    - SLT, SLTU
    - SLL, SRL, SRA
  - Interface:
    - input [31:0] A, B
    - input [3:0] op
    - input [4:0] shamt
    - output [31:0] hi, lo
    - output zero

# Arithmetic Logic Unit (ALU)



op	Function
0000	A and B
0001	A or B
0010	A nor B
0011	A xor B
0100	A + B
0101	A - B
0110	A * B (signed)
0111	A * B (unsigned)
1000	B << shamt
1001	B >> shamt
1010	B >>> shamt
1011	B >>> shamt
1100	A < B (signed)
1101	A < B (unsigned)
1110	A < B (unsigned)
1111	A < B (unsigned)



# Set Less Than

- Basic idea: set result to  
0000000000000000000000000000000000000001 if  $A < B$  else  
00
- To do this, compute  $\text{DIFF} = A - B$
- For signed SLT, look at sign bits of A, B, and DIFF
  - if A is negative and B is positive, set result to 1
  - if both positive and negative DIFF, set result to 1
    - Example:  $2 - 4 = -2$ ,  $4 - 2 = 2$
  - if both negative and negative DIFF, set result 1
    - Example  $-4 - (-2) = -2$ ,  $-2 - (-4) = 2$

# Set Less Than

- For unsigned, look at carryout

- Example for 4-bit operands:

4 = 0100 -> 0100 -> 0100  
-2 = 0010 -> 1101 -> + 1110  
0010 carryout = 1

2 = 0010 -> 0010 -> 0010  
- 4 = 0100 -> 1011 -> + 1100  
1110 carryout = 0

2 = 0010 -> 0010 -> 0010  
-2 = 0010 -> 1101 -> + 1110  
0000 carryout = 1

- If carryout = 0, A-B is < 0 therefore A < B



---

# Shifters

---

- **Logical shifter:** shifts value to left or right and fills empty spaces with 0's
  - Ex:  $11001 \gg 2 = 00110$
  - Ex:  $11001 \ll 2 = 00100$
- **Arithmetic shifter:** same as logical shifter, but on right shift, fills empty spaces with the old most significant bit (msb).
  - Ex:  $11001 \ggg 2 = 11110$
  - Ex:  $11001 \lll 2 = 00100$

---

# ALU Design 1

---

```
module alu (input [31:0] a,b,
            input[3:0] op,
            input [4:0] shamt,
            output reg [31:0] hi,lo,
            output zero);

wire [32:0] diff = {1'b0,a}+{1'b0,~b+32'b1};

assign zero = lo==32'd0 ? 1'b1 : 1'b0;

always @(*) begin
    hi = 32'b0;
    lo = 32'b0;
```





---

# ALU Design 2

---

```
casez (op)
    // arithmetic operations
    4'b01_00: lo = a+b; // add
    4'b01_01: lo = a-b; // sub

    // mult signed
    4'b01_10: {hi,lo} = $signed(a)*$signed(b);

    // mult unsigned
    4'b01_11: {hi,lo} = a*b;
```

---

# ALU Design 3

---

```
// shifter operations
4'b10_00: lo = b << shamt; // sll
4'b10_01: lo = b >> shamt; // srl
4'b10_1?: lo = $signed(b) >>> shamt; //sra
```



---

# ALU Design 4

---

```
// comparison operations
4'b11_00:
if (a[31] & ~b[31])
    lo = 32'b1; // a neg, b pos
else
    if (a[31] == b[31] & diff[31])
        // same sign, diff is neg
        lo = 32'b1;
    else lo = 32'b0;

4'b11_??:
if (~diff[32]) lo = 32'b1; // sltu
else lo = 32'b0;
```



---

# ALU Design 5

---

```
// logical operations
4'b00_00: lo = a & b; // and
4'b00_01: lo = a | b; // or
4'b00_10: lo = ~(a | b); // nor
4'b00_11: lo = a ^ b; // xor

endcase

end // always

endmodule
```

---

# ALU Design 5

---

```
module alu_reg (input clk, input[31:0] a,b, input[3:0] op, input [4:0] shamt, output
[31:0] hi,lo, output zero);

wire [31:0] a_reg,b_reg,hi_comb,lo_comb;
wire [3:0] op_reg;
wire [4:0] shamt_reg;
wire zero_comb;

regn #(32) input_reg_a(clk,a,a_reg);
regn #(32) input_reg_b(clk,b,b_reg);
regn #(4) input_reg_op(clk,op,op_reg);
regn #(5) input_reg_shamt(clk,shamt,shamt_reg);

alu my_alu(.a(a_reg), .b(b_reg), .shamt(shamt_reg), .op(op_reg), .hi(hi_comb),
.lo(lo_comb), .zero(zero_comb));

regn #(32) output_reg_hi(clk,hi_comb,hi);
regn #(32) output_reg_lo(clk,lo_comb,lo);
regn #(1) output_reg_zero(clk,zero_comb,zero);

endmodule
```



---

# Synthesis Results

---

- ALU\_reg requires:
  - 831/33216 LEs (3%)
  - 16 9-bit multipliers (signed and unsigned occur in parallel)
  - 94.95 MHz

---

# Lab 1

---

- Objective:
  - Create a self-checking testbench for the ALU
  - Create a test vectors file
  - Must cover all the test cases listed in the lab document
  - Make sure you describe any bugs you find in the design
- What to submit:
  - Test vector file
  - Corresponding file that describes each test vector
  - Testbench Verilog
  - Demo using SignalTap (using CLOCK\_50)
- Due: 9/30 by midnight